
 MODEL:
 ! A Simple Transportation Problem with soft constraints;
 ! Motivation: Sometimes a simple model is unexpectedly infeasible.
   How should one identify the cause of the unexpected infeasibility?
   A useful approach is that of making constraints "soft."
    Basic idea:  Add a "Superman" or artificial variable that can be used to always
   make the constraint feasible, however, the cost of the superman
   variable is very high, so that it will not be used if the 
   model really has a feasible solution;
 ! Keywords: Artificial, Debug, Distribution, Infeasibility, Infeasible, 
      LINGO, Shipping, Soft constraint, Transportation;
 SETS:
     SOURCE      : CAP;
     CUSTOMER    : DEM, SUPERSHIP;
     ROUTE( SOURCE, CUSTOMER)     : SCOST, SHIPT;
 ENDSETS
 ! Here are the parameters;
    DATA:
      SUPERCOST =  1000; ! Cost/unit of infeasibilities;
      SOURCE =  BILOXI LAX TAMPA; ! Names of supply points;
       CAP   =   30,    25,  15 ; ! Their capacities;
!  Names of demand points;
   CUSTOMER =   AURORA CASPER TONOPAH LUDLOW;
       DEM   =    24,    20,    95,    21 ;  ! Notice, the 95 for TONOPAH seems 
suspicious;
       SCOST =     6,     2,     6,     7,
                   4,     9,     5,     3,
                   8,     8,     1,     5 ;
    ENDDATA
!----------------------------------------------------;

SUBMODEL TRANIT:
! Variables:
    SHIPT( i, j) = amount shipped from source i to destination j,
    SUPERSHIP( j) = artificial amount shipped to j to achieve feasibility;
! The objective;
    MIN = OBJ;
    OBJ = COSTREAL +  SUPERCOST * SUPERVOL; ! Overall objective;
    COSTREAL = @SUM( ROUTE: SCOST * SHIPT); ! Cost of real shipments;
    SUPERVOL = @SUM( CUSTOMER( j): SUPERSHIP( J)); ! Cost of artificial shipments;
! The demand constraints;
    @FOR( CUSTOMER( J):
      [DEMRO]  @SUM( SOURCE( I): SHIPT( I, J)) + SUPERSHIP( j) = DEM( J));
! The supply constraints;
    @FOR( SOURCE( I):
      [CAPRO] @SUM( CUSTOMER( J): SHIPT( I, J)) <= CAP( I));
ENDSUBMODEL
CALC: 
  @SOLVE( TRANIT);
  @WRITE('          Real cost= ', COSTREAL, @NEWLINE(1));
  @WRITE(' Infeasibility cost= ', SUPERCOST * SUPERVOL, @NEWLINE(1));

! Display the real shipments;
  @FOR( ROUTE( I, J) | SHIPT( I, J) #GT# 0:
    @WRITE(' Ship ', @FORMAT( SHIPT(I,J),'5.1f'),' from ',
      @FORMAT( SOURCE(i),'7s'),' to ', @FORMAT( CUSTOMER(j),'7s'), @NEWLINE(1));
     );

  @WRITE(@NEWLINE( 1));  ! Display any artificial variables used;
  @FOR( CUSTOMER( J) | SUPERSHIP( J) #GT# 0:
    @WRITE(' Unsatisfied demand of  ', @FORMAT( SUPERSHIP( J),'5.1f'),'  at  ', @FORMAT( 
CUSTOMER(j),'7s'), @NEWLINE(1));
      );
ENDCALC
END


